NAG Library Routine Document

D03PKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D03PKF integrates a system of linear or nonlinear, first-order, time-dependent partial differential equations (PDEs) in one space variable, with scope for coupled ordinary differential equations (ODEs). The spatial discretization is performed using the Keller box scheme and the method of lines is employed to reduce the PDEs to a system of ODEs. The resulting system is solved using a Backward Differentiation Formula (BDF) method or a Theta method (switching between Newton's method and functional iteration).

2 Specification

```
 SUBROUTINE D03PKF(NPDE, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X, NLEFT,
1                  NCODE, ODEDEF, NXI, XI, NEQN, RTOL, ATOL, ITOL, NORM,
2                  LAOPT, ALGOPT, RSAVE, LRSAVE, ISAVE, LISAVE, ITASK,
3                  ITRACE, IND, IFAIL)

 INTEGER           NPDE, NPTS, NLEFT, NCODE, NXI, NEQN, ITOL, LRSAVE,
1                  ISAVE(LISAVE), LISAVE, ITASK, ITRACE, IND, IFAIL
 double precision  TS, TOUT, U(NEQN), X(NPTS), XI(*), RTOL(*), ATOL(*),
1                  ALGOPT(30), RSAVE(LRSAVE)
 CHARACTER*1       NORM, LAOPT
 EXTERNAL          PDEDEF, BNDARY, ODEDEF
```

3 Description

D03PKF integrates the system of first-order PDEs and coupled ODEs

$$G_i\left(x,t,U,U_x,U_t,V,\dot{V}\right)=0, \qquad i=1,2,\ldots,\text{NPDE}, \qquad a\le x\le b, t\ge t_0, \tag{1}$$

$$F_i\left(t,V,\dot{V},\xi,U^*,U_x^*,U_t^*\right)=0, \qquad i=1,2,\ldots,\text{NCODE}. \tag{2}$$

In the PDE part of the problem given by (1), the functions G_i must have the general form

$$G_i=\sum_{j=1}^{\text{NPDE}}P_{i,j}\frac{\partial U_j}{\partial t}+\sum_{j=1}^{\text{NCODE}}Q_{i,j}\dot{V}_j+R_i=0, \qquad i=1,2,\ldots,\text{NPDE}, \tag{3}$$

where $P_{i,j}$, $Q_{i,j}$ and R_i depend on x,t,U,U_x and V.

The vector U is the set of PDE solution values

$$U(x,t)=[U_1(x,t),\ldots,U_{\text{NPDE}}(x,t)]^{\text{T}},$$

and the vector U_x is the partial derivative with respect to x. The vector V is the set of ODE solution values

$$V(t)=[V_1(t),\ldots,V_{\text{NCODE}}(t)]^{\text{T}},$$

and $\dot{V}$ denotes its derivative with respect to time.

In the ODE part given by (2), ξ represents a vector of n_ξ spatial coupling points at which the ODEs are coupled to the PDEs. These points may or may not be equal to some of the PDE spatial mesh points. U^*, U_x^* and U_t^* are the functions U, U_x and U_t evaluated at these coupling points. Each F_i may only depend linearly on time derivatives. Hence equation (2) may be written more precisely as

$$F=A-B\dot{V}-CU_t^*, \tag{4}$$

where $F=[F_1,\ldots,F_{\text{NCODE}}]^{\text{T}}$, A is a vector of length NCODE, B is an NCODE by NCODE matrix, C is

an NCODE by $\left(n_\xi \times \text{NPDE}\right)$ matrix. The entries in A, B and C may depend on t, ξ, U^*, U_x^* and V. In practice you only need to supply a vector of information to define the ODEs and not the matrices B and C. (See Section 5 for the specification of ODEDEF.)

The integration in time is from t_0 to t_{out}, over the space interval $a \le x \le b$, where $a = x_1$ and $b = x_{\text{NPTS}}$ are the leftmost and rightmost points of a user-defined mesh $x_1, x_2, \ldots, x_{\text{NPTS}}$.

The PDE system which is defined by the functions G_i must be specified in PDEDEF.

The initial values of the functions $U(x,t)$ and $V(t)$ must be given at $t = t_0$.

For a first-order system of PDEs, only one boundary condition is required for each PDE component U_i. The NPDE boundary conditions are separated into n_a at the left-hand boundary $x = a$, and n_b at the right-hand boundary $x = b$, such that $n_a + n_b = \text{NPDE}$. The position of the boundary condition for each component should be chosen with care; the general rule is that if the characteristic direction of U_i at the left-hand boundary (say) points into the interior of the solution domain, then the boundary condition for U_i should be specified at the left-hand boundary. Incorrect positioning of boundary conditions generally results in initialization or integration difficulties in the underlying time integration routines.

The boundary conditions have the form:

$$G_i^L\left(x,t,U,U_t,V,\dot{V}\right) = 0 \qquad \text{at } x = a, \qquad i = 1,2,\ldots,n_a, \tag{5}$$

at the left-hand boundary, and

$$G_i^R\left(x,t,U,U_t,V,\dot{V}\right) = 0 \qquad \text{at } x = b, \qquad i = 1,2,\ldots,n_b, \tag{6}$$

at the right-hand boundary.

Note that the functions G_i^L and G_i^R must not depend on U_x, since spatial derivatives are not determined explicitly in the Keller box scheme. If the problem involves derivative (Neumann) boundary conditions then it is generally possible to restate such boundary conditions in terms of permissible variables. Also note that G_i^L and G_i^R must be linear with respect to time derivatives, so that the boundary conditions have the general form:

$$\sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} H_{i,j}^L \dot{V}_j + S_i^L = 0, \qquad i = 1,2,\ldots,n_a, \tag{7}$$

at the left-hand boundary, and

$$\sum_{j=1}^{\text{NPDE}} E_{i,j}^R \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} H_{i,j}^R \dot{V}_j + S_i^R = 0, \qquad i = 1,2,\ldots,n_b, \tag{8}$$

at the right-hand boundary, where $E_{i,j}^L$, $E_{i,j}^R$, $H_{i,j}^L$, $H_{i,j}^R$, S_i^L and S_i^R depend on x, t, U and V only.

The boundary conditions must be specified in BNDARY.

The problem is subject to the following restrictions:

(i) $P_{i,j}$, $Q_{i,j}$ and R_i must not depend on any time derivatives;

(ii) $t_0 < t_{\text{out}}$, so that integration is in the forward direction;

(iii) The evaluation of the function G_i is done approximately at the mid-points of the mesh $\text{X}(i)$, for $i = 1,2,\ldots,\text{NPTS}$, by calling the PDEDEF for each mid-point in turn. Any discontinuities in the function **must** therefore be at one or more of the mesh points $x_1, x_2, \ldots, x_{\text{NPTS}}$;

(iv) At least one of the functions $P_{i,j}$ must be nonzero so that there is a time derivative present in the PDE problem.

The algebraic-differential equation system which is defined by the functions F_i must be specified in ODEDEF. You must also specify the coupling points ξ in the array XI.

The parabolic equations are approximated by a system of ODEs in time for the values of U_i at mesh points. In this method of lines approach the Keller box scheme (see Keller (1970)) is applied to each PDE in the space variable only, resulting in a system of ODEs in time for the values of U_i at each mesh point.

In total there are NPDE $\times$ NPTS + NCODE ODEs in time direction. This system is then integrated forwards in time using a Backward Differentiation Formula (BDF) or a Theta method.

4 References

Berzins M (1990) Developments in the NAG Library software for parabolic equations *Scientific Software Systems* (eds J C Mason and M G Cox) 59–72 Chapman and Hall

Berzins M, Dew P M and Furzeland R M (1989) Developing software for time-dependent problems using the method of lines and differential-algebraic integrators *Appl. Numer. Math.* **5** 375–397

Berzins M and Furzeland R M (1992) An adaptive theta method for the solution of stiff and nonstiff differential equations *Appl. Numer. Math.* **9** 1–19

Keller H B (1970) A new difference scheme for parabolic problems *Numerical Solutions of Partial Differential Equations* (ed J Bramble) **2** 327–350 Academic Press

Pennington S V and Berzins M (1994) New NAG Library software for first-order partial differential equations *ACM Trans. Math. Softw.* **20** 63–99

5 Parameters

1: NPDE – INTEGER *Input*

On entry: the number of PDEs to be solved.

Constraint: NPDE ≥ 1.

2: TS – ***double precision*** *Input/Output*

On entry: the initial value of the independent variable t.

Constraint: TS $<$ TOUT.

On exit: the value of t corresponding to the solution in U. Normally TS $=$ TOUT.

3: TOUT – ***double precision*** *Input*

On entry: the final value of t to which the integration is to be carried out.

4: PDEDEF – SUBROUTINE, supplied by the user. *External Procedure*

PDEDEF must evaluate the functions G_i which define the system of PDEs. PDEDEF is called approximately midway between each pair of mesh points in turn by D03PKF.

The specification of PDEDEF is:

```
SUBROUTINE PDEDEF(NPDE, T, X, U, UT, UX, NCODE, V, VDOT, RES, IRES)
INTEGER           NPDE, NCODE, IRES
double precision  T, X, U(NPDE), UT(NPDE), UX(NPDE), V(NCODE),
1                 VDOT(NCODE), RES(NPDE)
```

1: NPDE – INTEGER *Input*

On entry: the number of PDEs in the system.

2: T – ***double precision*** *Input*

On entry: the current value of the independent variable t.

3: X – ***double precision*** *Input*

On entry: the current value of the space variable x.

4: U(NPDE) – ***double precision*** array *Input*

On entry: U(i) contains the value of the component $U_i(x,t)$, for $i = 1,2,\ldots,$NPDE.

5: UT(NPDE) – ***double precision*** array *Input*

On entry: UT(i) contains the value of the component $\dfrac{\partial U_i(x,t)}{\partial t}$, for $i = 1,2,\ldots,$NPDE.

6: UX(NPDE) – ***double precision*** array *Input*

On entry: UX(i) contains the value of the component $\dfrac{\partial U_i(x,t)}{\partial x}$, for $i = 1,2,\ldots,$NPDE.

7: NCODE – INTEGER *Input*

On entry: the number of coupled ODEs in the system.

8: V(NCODE) – ***double precision*** array *Input*

On entry: if NCODE > 0, V(i) contains the value of component $V_i(t)$, for $i = 1,2,\ldots,$NCODE.

9: VDOT(NCODE) – ***double precision*** array *Input*

On entry: if NCODE > 0, VDOT(i) contains the value of component $\dot{V}_i(t)$, for $i = 1,2,\ldots,$NCODE.

10: RES(NPDE) – ***double precision*** array *Output*

On exit: RES(i) must contain the ith component of G, for $i = 1,2,\ldots,$NPDE, where G is defined as

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j}\frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} Q_{i,j}\dot{V}_j, \tag{9}$$

i.e., only terms depending explicitly on time derivatives, or

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j}\frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} Q_{i,j}\dot{V}_j + R_i, \tag{10}$$

i.e., all terms in equation (3).

The definition of G is determined by the input value of IRES.

11: IRES – INTEGER *Input/Output*

On entry: the form of G_i that must be returned in the array RES.

IRES $= -1$

Equation (9) must be used.

IRES $= 1$

Equation (10) must be used.

On exit: should usually remain unchanged. However, you may set IRES to force the integration routine to take certain actions, as described below:

IRES $= 2$

Indicates to the integrator that control should be passed back immediately to the calling subroutine with the error indicator set to IFAIL $= 6$.

IRES $= 3$

Indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. You may wish to set IRES $= 3$ when a physically

meaningless input or output value has been generated. If you consecutively set IRES = 3, then D03PKF returns to the calling subroutine with the error indicator set to IFAIL = 4.

PDEDEF must be declared as EXTERNAL in the (sub)program from which D03PKF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

5: BNDARY – SUBROUTINE, supplied by the user. *External Procedure*

BNDARY must evaluate the functions G_i^L and G_i^R which describe the boundary conditions, as given in (5) and (6).

The specification of BNDARY is:

```
SUBROUTINE BNDARY(NPDE, T, IBND, NOBC, U, UT, NCODE, V, VDOT, RES,
1                 IRES)
INTEGER           NPDE, IBND, NOBC, NCODE, IRES
double precision  T, U(NPDE), UT(NPDE), V(NCODE), VDOT(NCODE),
1                 RES(NOBC)
```

1: NPDE – INTEGER *Input*

On entry: the number of PDEs in the system.

2: T – ***double precision*** *Input*

On entry: the current value of the independent variable t.

3: IBND – INTEGER *Input*

On entry: specifies which boundary conditions are to be evaluated.

IBND $= 0$
BNDARY must compute the left-hand boundary condition at $x = a$.

IBND $\neq 0$
BNDARY must compute the right-hand boundary condition at $x = b$.

4: NOBC – INTEGER *Input*

On entry: specifies the number of boundary conditions at the boundary specified by IBND.

5: U(NPDE) – ***double precision*** array *Input*

On entry: U(i) contains the value of the component $U_i(x,t)$ at the boundary specified by IBND, for $i = 1, 2, \ldots,$ NPDE.

6: UT(NPDE) – ***double precision*** array *Input*

On entry: UT(i) contains the value of the component $\dfrac{\partial U_i(x,t)}{\partial t}$ at the boundary specified by IBND, for $i = 1, 2, \ldots,$ NPDE.

7: NCODE – INTEGER *Input*

On entry: the number of coupled ODEs in the system.

8: V(NCODE) – ***double precision*** array *Input*

On entry: if NCODE > 0, V(i) contains the value of component $V_i(t)$, for $i = 1, 2, \ldots,$ NCODE.

9: VDOT(NCODE) – ***double precision*** array *Input*

On entry: if NCODE > 0, VDOT(i) contains the value of component $\dot{V}_i(t)$, for $i = 1, 2, \ldots,$ NCODE.

Note: VDOT(i), for $i = 1, 2, \ldots,$ NCODE, may only appear linearly as in (7) and (8).

10: RES(NOBC) – ***double precision*** array *Output*

On exit: RES(i) must contain the ith component of G^L or G^R, depending on the value of IBND, for $i = 1, 2, \ldots,$ NOBC, where G^L is defined as

$$G_i^L = \sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} H_{i,j}^L \dot{V}_j, \tag{11}$$

i.e., only terms depending explicitly on time derivatives, or

$$G_i^L = \sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + \sum_{j=1}^{\text{NCODE}} H_{i,j}^L \dot{V}_j + S_i^L, \tag{12}$$

i.e., all terms in equation (7), and similarly for G_i^R.

The definitions of G^L and G^R are determined by the input value of IRES.

11: IRES – INTEGER *Input/Output*

On entry: the form of G_i^L (or G_i^R) that must be returned in the array RES.

IRES $= -1$
Equation (11) must be used.

IRES $= 1$
Equation (12) must be used.

On exit: should usually remain unchanged. However, you may set IRES to force the integration routine to take certain actions, as described below:

IRES $= 2$
Indicates to the integrator that control should be passed back immediately to the calling subroutine with the error indicator set to IFAIL $= 6$.

IRES $= 3$
Indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. You may wish to set IRES $= 3$ when a physically meaningless input or output value has been generated. If you consecutively set IRES $= 3$, then D03PKF returns to the calling subroutine with the error indicator set to IFAIL $= 4$.

BNDARY must be declared as EXTERNAL in the (sub)program from which D03PKF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

6: U(NEQN) – ***double precision*** array *Input/Output*

On entry: the initial values of the dependent variables defined as follows:

U(NPDE $\times (j-1) + i$) contain $U_i(x_j, t_0)$, for $i = 1, 2, \ldots,$ NPDE and $j = 1, 2, \ldots,$ NPTS, and
U(NPTS $\times$ NPDE $+ i$) contain $V_i(t_0)$, for $i = 1, 2, \ldots,$ NCODE.

On exit: the computed solution $U_i(x_j, t)$, for $i = 1, 2, \ldots,$ NPDE and $j = 1, 2, \ldots,$ NPTS, and $V_k(t)$, for $k = 1, 2, \ldots,$ NCODE, evaluated at $t =$ TS.

7: NPTS – INTEGER *Input*

On entry: the number of mesh points in the interval $[a, b]$.

Constraint: NPTS ≥ 3.

8: X(NPTS) – ***double precision*** array *Input*

On entry: the mesh points in the space direction. X(1) must specify the left-hand boundary, a, and X(NPTS) must specify the right-hand boundary, b.

Constraint: $\mathrm{X}(1) < \mathrm{X}(2) < \cdots < \mathrm{X}(\mathrm{NPTS})$.

9: NLEFT – INTEGER *Input*

On entry: the number n_a of boundary conditions at the left-hand mesh point X(1).

Constraint: $0 \leq \mathrm{NLEFT} \leq \mathrm{NPDE}$.

10: NCODE – INTEGER *Input*

On entry: the number of coupled ODE components.

Constraint: NCODE ≥ 0.

11: ODEDEF – SUBROUTINE, supplied by the NAG Library or the user. *External Procedure*

ODEDEF must evaluate the functions F, which define the system of ODEs, as given in (4). If you wish to compute the solution of a system of PDEs only (i.e., NCODE = 0), ODEDEF must be the dummy routine D03PEK. D03PEK is included in the NAG Library.

The specification of ODEDEF is:

```
      SUBROUTINE ODEDEF(NPDE, T, NCODE, V, VDOT, NXI, XI, UCP, UCPX, UCPT,
     1                  F, IRES)
      INTEGER            NPDE, NCODE, NXI, IRES
      double precision   T, V(NCODE), VDOT(NCODE), XI(NXI), UCP(NPDE,*),
     1                   UCPX(NPDE,*), UCPT(NPDE,*), F(NCODE)
```

1: NPDE – INTEGER *Input*

On entry: the number of PDEs in the system.

2: T – ***double precision*** *Input*

On entry: the current value of the independent variable t.

3: NCODE – INTEGER *Input*

On entry: the number of coupled ODEs in the system.

4: V(NCODE) – ***double precision*** array *Input*

On entry: if NCODE > 0, V(i) contains the value of component $V_i(t)$, for $i = 1, 2, \ldots, \mathrm{NCODE}$.

5: VDOT(NCODE) – ***double precision*** array *Input*

On entry: if NCODE > 0, VDOT(i) contains the value of component $\dot{V}_i(t)$, for $i = 1, 2, \ldots, \mathrm{NCODE}$.

6: NXI – INTEGER *Input*

On entry: the number of ODE/PDE coupling points.

7: XI(NXI) – ***double precision*** array *Input*

On entry: if NXI > 0, XI(i) contains the ODE/PDE coupling points, ξ_i, for $i = 1, 2, \ldots, \text{NXI}$.

8: UCP(NPDE,$*$) – ***double precision*** array *Input*

Note: the second dimension of the array UCP is $\max(1, \text{NXI})$.

On entry: if NXI > 0, UCP(i, j) contains the value of $U_i(x, t)$ at the coupling point $x = \xi_j$, for $i = 1, 2, \ldots, \text{NPDE}$ and $j = 1, 2, \ldots, \text{NXI}$.

9: UCPX(NPDE,$*$) – ***double precision*** array *Input*

Note: the second dimension of the array UCPX is $\max(1, \text{NXI})$.

On entry: if NXI > 0, UCPX(i, j) contains the value of $\dfrac{\partial U_i(x, t)}{\partial x}$ at the coupling point $x = \xi_j$, for $i = 1, 2, \ldots, \text{NPDE}$ and $j = 1, 2, \ldots, \text{NXI}$.

10: UCPT(NPDE,$*$) – ***double precision*** array *Input*

Note: the second dimension of the array UCPT is $\max(1, \text{NXI})$.

On entry: if NXI > 0, UCPT(i, j) contains the value of $\dfrac{\partial U_i}{\partial t}$ at the coupling point $x = \xi_j$, for $i = 1, 2, \ldots, \text{NPDE}$ and $j = 1, 2, \ldots, \text{NXI}$.

11: F(NCODE) – ***double precision*** array *Output*

On exit: if NCODE > 0, F(i) must contain the ith component of F, for $i = 1, 2, \ldots, \text{NCODE}$, where F is defined as

$$F = -B\dot{V} - CU_t^*, \tag{13}$$

i.e., only terms depending explicitly on time derivatives, or

$$F = A - B\dot{V} - CU_t^*, \tag{14}$$

i.e., all terms in equation (4). The definition of F is determined by the input value of IRES.

12: IRES – INTEGER *Input/Output*

On entry: the form of F that must be returned in the array F.

IRES $= -1$

Equation (13) must be used.

IRES $= 1$

Equation (14) must be used.

On exit: should usually remain unchanged. However, you may reset IRES to force the integration routine to take certain actions, as described below:

IRES $= 2$

Indicates to the integrator that control should be passed back immediately to the calling subroutine with the error indicator set to IFAIL $= 6$.

IRES $= 3$

Indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. You may wish to set IRES $= 3$ when a physically meaningless input or output value has been generated. If you consecutively set IRES $= 3$, then D03PKF returns to the calling subroutine with the error indicator set to IFAIL $= 4$.

ODEDEF must be declared as EXTERNAL in the (sub)program from which D03PKF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

12: NXI – INTEGER *Input*

On entry: the number of ODE/PDE coupling points.

Constraints:

if NCODE $= 0$, NXI $= 0$;
if NCODE > 0, NXI ≥ 0.

13: XI$(*)$ – ***double precision*** array *Input*

Note: the dimension of the array XI must be at least $\max(1, \text{NXI})$.

On entry: XI(i), for $i = 1, 2, \ldots, \text{NXI}$, must be set to the ODE/PDE coupling points, ξ_i.

Constraint: $\text{X}(1) \leq \text{XI}(1) < \text{XI}(2) < \cdots < \text{XI}(\text{NXI}) \leq \text{X}(\text{NPTS})$.

14: NEQN – INTEGER *Input*

On entry: the number of ODEs in the time direction.

Constraint: NEQN $=$ NPDE $\times$ NPTS $+$ NCODE.

15: RTOL$(*)$ – ***double precision*** array *Input*

Note: the dimension of the array RTOL must be at least 1 if ITOL $= 1$ or 2 and at least NEQN if ITOL $= 3$ or 4.

On entry: the relative local error tolerance.

Constraint: RTOL$(i) \geq 0$ for all relevant i.

16: ATOL$(*)$ – ***double precision*** array *Input*

Note: the dimension of the array ATOL must be at least 1 if ITOL $= 1$ or 3 and at least NEQN if ITOL $= 2$ or 4.

On entry: the absolute local error tolerance.

Constraint: ATOL$(i) \geq 0$ for all relevant i.

Note: corresponding elements of RTOL and ATOL cannot both be 0.0.

17: ITOL – INTEGER *Input*

On entry: a value to indicate the form of the local error test. ITOL indicates to D03PKF whether to interpret either or both of RTOL or ATOL as a vector or scalar. The error test to be satisfied is $\|e_i/w_i\| < 1.0$, where w_i is defined as follows:

ITOL	RTOL	ATOL	w_i
1	scalar	scalar	$\text{RTOL}(1) \times \lvert\text{U}(i)\rvert + \text{ATOL}(1)$
2	scalar	vector	$\text{RTOL}(1) \times \lvert\text{U}(i)\rvert + \text{ATOL}(i)$
3	vector	scalar	$\text{RTOL}(i) \times \lvert\text{U}(i)\rvert + \text{ATOL}(1)$
4	vector	vector	$\text{RTOL}(i) \times \lvert\text{U}(i)\rvert + \text{ATOL}(i)$

In the above, e_i denotes the estimated local error for the ith component of the coupled PDE/ODE system in time, U(i), for $i = 1, 2, \ldots, \text{NEQN}$.

The choice of norm used is defined by the parameter NORM.

Constraint: $1 \leq \text{ITOL} \leq 4$.

18: NORM – CHARACTER*1 *Input*

On entry: the type of norm to be used.

NORM = 'M'
Maximum norm.

NORM = 'A'
Averaged L_2 norm.

If $\mathrm{U}_{\mathrm{norm}}$ denotes the norm of the vector U of length NEQN, then for the averaged L_2 norm

$$\mathrm{U}_{\mathrm{norm}} = \sqrt{\frac{1}{\mathrm{NEQN}} \sum_{i=1}^{\mathrm{NEQN}} (\mathrm{U}(i)/w_i)^2},$$

while for the maximum norm

$$\mathrm{U}_{\mathrm{norm}} = \max_i |\mathrm{U}(i)/w_i|.$$

See the description of ITOL for the formulation of the weight vector w.

Constraint: NORM = 'M' or 'A'.

19: LAOPT – CHARACTER*1 *Input*

On entry: the type of matrix algebra required.

LAOPT = 'F'
Full matrix methods to be used.

LAOPT = 'B'
Banded matrix methods to be used.

LAOPT = 'S'
Sparse matrix methods to be used.

Constraint: LAOPT = 'F', 'B' or 'S'.

Note: you are recommended to use the banded option when no coupled ODEs are present (i.e., NCODE = 0).

20: ALGOPT(30) – ***double precision*** array *Input*

On entry: may be set to control various options available in the integrator. If you wish to employ all the default options, then ALGOPT(1) should be set to 0.0. Default values will also be used for any other elements of ALGOPT set to zero. The permissible values, default values, and meanings are as follows:

ALGOPT(1)
Selects the ODE integration method to be used. If ALGOPT(1) = 1.0, a BDF method is used and if ALGOPT(1) = 2.0, a Theta method is used. The default value is ALGOPT(1) = 1.0.

If ALGOPT(1) = 2.0, then ALGOPT(i), for $i = 2, 3, 4$ are not used.

ALGOPT(2)
Specifies the maximum order of the BDF integration formula to be used. ALGOPT(2) may be 1.0, 2.0, 3.0, 4.0 or 5.0. The default value is ALGOPT(2) = 5.0.

ALGOPT(3)
Specifies what method is to be used to solve the system of nonlinear equations arising on each step of the BDF method. If ALGOPT(3) = 1.0 a modified Newton iteration is used and if ALGOPT(3) = 2.0 a functional iteration method is used. If functional iteration is selected and the integrator encounters difficulty, then there is an automatic switch to the modified Newton iteration. The default value is ALGOPT(3) = 1.0.

ALGOPT(4)
: Specifies whether or not the Petzold error test is to be employed. The Petzold error test results in extra overhead but is more suitable when algebraic equations are present, such as $P_{i,j} = 0.0$, for $j = 1, 2, \ldots,$ NPDE for some i or when there is no $\dot{V}_i(t)$ dependence in the coupled ODE system. If ALGOPT(4) $= 1.0$, then the Petzold test is used. If ALGOPT(4) $= 2.0$, then the Petzold test is not used. The default value is ALGOPT(4) $= 1.0$.

If ALGOPT(1) $= 1.0$, then ALGOPT(i), for $i = 5, 6, 7$ are not used.

ALGOPT(5)
: Specifies the value of Theta to be used in the Theta integration method. $0.51 \leq$ ALGOPT(5) ≤ 0.99. The default value is ALGOPT(5) $= 0.55$.

ALGOPT(6)
: Specifies what method is to be used to solve the system of nonlinear equations arising on each step of the Theta method. If ALGOPT(6) $= 1.0$, a modified Newton iteration is used and if ALGOPT(6) $= 2.0$, a functional iteration method is used. The default value is ALGOPT(6) $= 1.0$.

ALGOPT(7)
: Specifies whether or not the integrator is allowed to switch automatically between modified Newton and functional iteration methods in order to be more efficient. If ALGOPT(7) $= 1.0$, then switching is allowed and if ALGOPT(7) $= 2.0$, then switching is not allowed. The default value is ALGOPT(7) $= 1.0$.

ALGOPT(11)
: Specifies a point in the time direction, t_{crit}, beyond which integration must not be attempted. The use of t_{crit} is described under the parameter ITASK. If ALGOPT(1) $\neq 0.0$, a value of 0.0, for ALGOPT(11), say, should be specified even if ITASK subsequently specifies that t_{crit} will not be used.

ALGOPT(12)
: Specifies the minimum absolute step size to be allowed in the time integration. If this option is not required, ALGOPT(12) should be set to 0.0.

ALGOPT(13)
: Specifies the maximum absolute step size to be allowed in the time integration. If this option is not required, ALGOPT(13) should be set to 0.0.

ALGOPT(14)
: Specifies the initial step size to be attempted by the integrator. If ALGOPT(14) $= 0.0$, then the initial step size is calculated internally.

ALGOPT(15)
: Specifies the maximum number of steps to be attempted by the integrator in any one call. If ALGOPT(15) $= 0.0$, then no limit is imposed.

ALGOPT(23)
: Specifies what method is to be used to solve the nonlinear equations at the initial point to initialize the values of U, U_t, V and $\dot{V}$. If ALGOPT(23) $= 1.0$, a modified Newton iteration is used and if ALGOPT(23) $= 2.0$, functional iteration is used. The default value is ALGOPT(23) $= 1.0$.

ALGOPT(29) and ALGOPT(30) are used only for the sparse matrix algebra option, i.e., LAOPT = 'S'.

ALGOPT(29)
: Governs the choice of pivots during the decomposition of the first Jacobian matrix. It should lie in the range $0.0 <$ ALGOPT(29) < 1.0, with smaller values biasing the algorithm towards maintaining sparsity at the expense of numerical stability. If ALGOPT(29) lies outside this range then the default value is used. If the routines regard the Jacobian matrix as numerically singular then increasing ALGOPT(29) towards 1.0 may help, but at the cost of increased fill-in. The default value is ALGOPT(29) $= 0.1$.

ALGOPT(30)
Used as a relative pivot threshold during subsequent Jacobian decompositions (see ALGOPT(29)) below which an internal error is invoked. ALGOPT(30) must be greater than zero, otherwise the default value is used. If ALGOPT(30) is greater than 1.0 no check is made on the pivot size, and this may be a necessary option if the Jacobian is found to be numerically singular (see ALGOPT(29)). The default value is ALGOPT(30) = 0.0001.

21: RSAVE(LRSAVE) – ***double precision*** array *Communication Array*

If IND = 0, RSAVE need not be set on entry.

If IND = 1, RSAVE must be unchanged from the previous call to the routine because it contains required information about the iteration.

22: LRSAVE – INTEGER *Input*

On entry: the dimension of the array RSAVE as declared in the (sub)program from which D03PKF is called. Its size depends on the type of matrix algebra selected.

If LAOPT = 'F', LRSAVE $\geq$ NEQN $\times$ NEQN + NEQN + $nwkres + lenode$.

If LAOPT = 'B', LRSAVE $\geq (2 \times ml + mu + 2) \times$ NEQN + $nwkres + lenode$.

If LAOPT = 'S', LRSAVE $\geq 4 \times$ NEQN + 11 $\times$ NEQN/2 + 1 + $nwkres + lenode$.

Where

ml and mu are the lower and upper half bandwidths given by $ml =$ NPDE + NLEFT $-$ 1 such that $mu = 2 \times$ NPDE $-$ NLEFT $-$ 1, for problems involving PDEs only; or
$ml = mu =$ NEQN $-$ 1, for coupled PDE/ODE problems.

$$nwkres = \begin{cases} \text{NPDE} \times (3 \times \text{NPDE} + 6 \times \text{NXI} + \text{NPTS} + 15) + \text{NXI} + \text{NCODE} + 7 \times \text{NPTS} + 2, & \text{when NCODE} > 0 \text{ and NXI} > 0; \text{ or} \\ \text{NPDE} \times (3 \times \text{NPDE} + \text{NPTS} + 21) + \text{NCODE} + 7 \times \text{NPTS} + 3, & \text{when NCODE} > 0 \text{ and NXI} = 0; \text{ or} \\ \text{NPDE} \times (3 \times \text{NPDE} + \text{NPTS} + 21) + 7 \times \text{NPTS} + 4, & \text{when NCODE} = 0. \end{cases}$$

$$lenode = \begin{cases} (6 + \text{int}(\text{ALGOPT}(2))) \times \text{NEQN} + 50, & \text{when the BDF method is used; or} \\ 9 \times \text{NEQN} + 50, & \text{when the Theta method is used.} \end{cases}$$

Note: when using the sparse option, the value of LRSAVE may be too small when supplied to the integrator. An estimate of the minimum size of LRSAVE is printed on the current error message unit if ITRACE > 0 and the routine returns with IFAIL = 15.

23: ISAVE(LISAVE) – INTEGER array *Communication Array*

If IND = 0, ISAVE need not be set.

If IND = 1, ISAVE must be unchanged from the previous call to the routine because it contains required information about the iteration. In particular the following components of the array ISAVE concern the efficiency of the integration:

ISAVE(1)
Contains the number of steps taken in time.

ISAVE(2)
Contains the number of residual evaluations of the resulting ODE system used. One such evaluation involves evaluating the PDE functions at all the mesh points, as well as one evaluation of the functions in the boundary conditions.

ISAVE(3)
Contains the number of Jacobian evaluations performed by the time integrator.

ISAVE(4)
Contains the order of the ODE method last used in the time integration.

ISAVE(5)
Contains the number of Newton iterations performed by the time integrator. Each iteration involves residual evaluation of the resulting ODE system followed by a back-substitution using the LU decomposition of the Jacobian matrix.

24: LISAVE – INTEGER *Input*

On entry: the dimension of the array ISAVE as declared in the (sub)program from which D03PKF is called. Its size depends on the type of matrix algebra selected:

if LAOPT = 'F', LISAVE $\geq$ 24;

if LAOPT = 'B', LISAVE $\geq$ NEQN + 24;

if LAOPT = 'S', LISAVE $\geq 25 \times$ NEQN + 24.

Note: when using the sparse option, the value of LISAVE may be too small when supplied to the integrator. An estimate of the minimum size of LISAVE is printed on the current error message unit if ITRACE > 0 and the routine returns with IFAIL = 15.

25: ITASK – INTEGER *Input*

On entry: the task to be performed by the ODE integrator.

ITASK = 1
Normal computation of output values U at $t =$ TOUT (by overshooting and interpolating).

ITASK = 2
Take one step in the time direction and return.

ITASK = 3
Stop at first internal integration point at or beyond $t =$ TOUT.

ITASK = 4
Normal computation of output values U at $t =$ TOUT but without overshooting $t = t_{\text{crit}}$, where t_{crit} is described under the parameter ALGOPT.

ITASK = 5
Take one step in the time direction and return, without passing t_{crit}, where t_{crit} is described under the parameter ALGOPT.

Constraint: ITASK = 1, 2, 3, 4 or 5.

26: ITRACE – INTEGER *Input*

On entry: the level of trace information required from D03PKF and the underlying ODE solver as follows:

ITRACE ≤ -1
No output is generated.

ITRACE = 0
Only warning messages from the PDE solver are printed on the current error message unit (see X04AAF).

ITRACE = 1
Output from the underlying ODE solver is printed on the current advisory message unit (see X04ABF). This output contains details of Jacobian entries, the nonlinear iteration and the time integration during the computation of the ODE system.

ITRACE = 2
Output from the underlying ODE solver is similar to that produced when ITRACE = 1, except that the advisory messages are given in greater detail.

ITRACE $\geq$ 3
Output from the underlying ODE solver is similar to that produced when ITRACE = 2, except that the advisory messages are given in greater detail.

You advised to set ITRACE = 0, unless you are experienced with sub-chapter D02M–N.

27: IND – INTEGER *Input/Output*

On entry: indicates whether this is a continuation call or a new integration.

IND $= 0$
Starts or restarts the integration in time.

IND $= 1$
Continues the integration after an earlier exit from the routine. In this case, only the parameters TOUT and IFAIL should be reset between calls to D03PKF.

Constraint: IND $= 0$ or 1.

On exit: IND $= 1$.

28: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, (TOUT − TS) is too small,
or ITASK $\neq 1$, 2, 3, 4 or 5,
or at least one of the coupling points defined in array XI is outside the interval [X(1), X(NPTS)],
or NPTS < 3,
or NPDE < 1,
or NLEFT not in range 0 to NPDE,
or NORM $\neq$ 'A' or 'M',
or LAOPT $\neq$ 'F', 'B' or 'S',
or ITOL $\neq 1$, 2, 3 or 4,
or IND $\neq 0$ or 1,
or mesh points X(i) are badly ordered,
or LRSAVE or LISAVE are too small,
or NCODE and NXI are incorrectly defined,
or IND $= 1$ on initial entry to D03PKF,
or an element of RTOL or ATOL < 0.0,
or corresponding elements of ATOL and RTOL are both 0.0,
or NEQN $\neq$ NPDE $\times$ NPTS + NCODE.

IFAIL $= 2$

The underlying ODE solver cannot make any further progress, with the values of ATOL and RTOL, across the integration range from the current point $t =$ TS. The components of U contain the computed values at the current point $t =$ TS.

IFAIL $= 3$

In the underlying ODE solver, there were repeated error test failures on an attempted step, before completing the requested task, but the integration was successful as far as $t =$ TS. The problem may have a singularity, or the error requirement may be inappropriate. Incorrect positioning of boundary conditions may also result in this error.

IFAIL $= 4$

In setting up the ODE system, the internal initialization routine was unable to initialize the derivative of the ODE system. This could be due to the fact that IRES was repeatedly set to 3 in one of PDEDEF, BNDARY or ODEDEF, when the residual in the underlying ODE solver was being evaluated. Incorrect positioning of boundary conditions may also result in this error.

IFAIL $= 5$

In solving the ODE system, a singular Jacobian has been encountered. You should check their problem formulation.

IFAIL $= 6$

When evaluating the residual in solving the ODE system, IRES was set to 2 in one of PDEDEF, BNDARY or ODEDEF. Integration was successful as far as $t =$ TS.

IFAIL $= 7$

The values of ATOL and RTOL are so small that the routine is unable to start the integration in time.

IFAIL $= 8$

In either, PDEDEF, BNDARY or ODEDEF, IRES was set to an invalid value.

IFAIL $= 9$ (D02NNF)

A serious error has occurred in an internal call to the specified routine. Check the problem specification and all parameters and array dimensions. Setting ITRACE $= 1$ may provide more information. If the problem persists, contact NAG.

IFAIL $= 10$

The required task has been completed, but it is estimated that a small change in ATOL and RTOL is unlikely to produce any change in the computed solution. (Only applies when you are not operating in one step mode, that is when ITASK $\neq 2$ or 5.)

IFAIL $= 11$

An error occurred during Jacobian formulation of the ODE system (a more detailed error description may be directed to the current advisory message unit). If using the sparse matrix algebra option, the values of ALGOPT(29) and ALGOPT(30) may be inappropriate.

IFAIL $= 12$

In solving the ODE system, the maximum number of steps specified in ALGOPT(15) has been taken.

IFAIL $= 13$

Some error weights w_i became zero during the time integration (see the description of ITOL). Pure relative error control (ATOL$(i) = 0.0$) was requested on a variable (the ith) which has become zero. The integration was successful as far as $t =$ TS.

IFAIL $= 14$

Not applicable.

IFAIL = 15

When using the sparse option, the value of LISAVE or LRSAVE was insufficient (more detailed information may be directed to the current error message unit).

7 Accuracy

D03PKF controls the accuracy of the integration in the time direction but not the accuracy of the approximation in space. The spatial accuracy depends on both the number of mesh points and on their distribution in space. In the time integration only the local error over a single step is controlled and so the accuracy over a number of steps cannot be guaranteed. You should therefore test the effect of varying the accuracy parameters, ATOL and RTOL.

8 Further Comments

The Keller box scheme can be used to solve higher-order problems which have been reduced to first-order by the introduction of new variables (see the example in Section 9). In general, a second-order problem can be solved with slightly greater accuracy using the Keller box scheme instead of a finite-difference scheme (see D03PCF/D03PCA or D03PHF/D03PHA for example), but at the expense of increased CPU time due to the larger number of function evaluations required.

It should be noted that the Keller box scheme, in common with other central-difference schemes, may be unsuitable for some hyperbolic first-order problems such as the apparently simple linear advection equation $U_t + aU_x = 0$, where a is a constant, resulting in spurious oscillations due to the lack of dissipation. This type of problem requires a discretization scheme with upwind weighting (D03PLF for example), or the addition of a second-order artificial dissipation term.

The time taken depends on the complexity of the system and on the accuracy requested. For a given system and a fixed accuracy it is approximately proportional to NEQN.

9 Example

This example provides a simple coupled system of two PDEs and one ODE.

$$(V_1)^2\frac{\partial U_1}{\partial t} - xV_1\dot{V}_1U_2 - \frac{\partial U_2}{\partial x} = 0,$$

$$U_2 - \frac{\partial U_1}{\partial x} = 0,$$

$$\dot{V}_1 - V_1U_1 - U_2 - 1 - t = 0,$$

for $t \in \left[10^{-4}, 0.1 \times 2^i\right]$, for $i = 1, 2, \ldots, 5, x \in [0, 1]$. The left boundary condition at $x = 0$ is

$$U_2 = -V_1 \exp t,$$

and the right boundary condition at $x = 1$ is

$$U_2 = -V_1\dot{V}_1.$$

The initial conditions at $t = 10^{-4}$ are defined by the exact solution:

$$V_1 = t, U_1(x,t) = \exp\{t(1-x)\} - 1.0 \qquad \text{and} \qquad U_2(x,t) = -t\exp\{t(1-x)\},\ x \in [0,1],$$

and the coupling point is at $\xi_1 = 1.0$.

This problem is exactly the same as the D03PHF/D03PHA example problem, but reduced to first-order by the introduction of a second PDE variable (as mentioned in Section 8).

9.1 Program Text

```
*     D03PKF Example Program Text
*     Mark 16 Release. NAG Copyright 1993.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NPDE, NPTS, NCODE, NXI, NLEFT, NEQN, LISAVE,
     +                 NWKRES, LENODE, LRSAVE
      PARAMETER        (NPDE=2,NPTS=21,NCODE=1,NXI=1,NLEFT=1,
     +                 NEQN=NPDE*NPTS+NCODE,LISAVE=24,
     +                 NWKRES=NPDE*(NPTS+6*NXI+3*NPDE+15)
     +                 +NCODE+NXI+7*NPTS+2,LENODE=11*NEQN+50,
     +                 LRSAVE=NEQN*NEQN+NEQN+NWKRES+LENODE)
*     .. Scalars in Common ..
      DOUBLE PRECISION TS
*     .. Local Scalars ..
      DOUBLE PRECISION TOUT
      INTEGER          I, IFAIL, IND, IT, ITASK, ITOL, ITRACE
      LOGICAL          THETA
      CHARACTER        LAOPT, NORM
*     .. Local Arrays ..
      DOUBLE PRECISION ALGOPT(30), ATOL(1), EXY(NEQN), RSAVE(LRSAVE),
     +                 RTOL(1), U(NEQN), X(NPTS), XI(1)
      INTEGER          ISAVE(LISAVE)
*     .. External Subroutines ..
      EXTERNAL         BNDARY, D03PKF, EXACT, ODEDEF, PDEDEF, UVINIT
*     .. Common blocks ..
      COMMON           /TAXIS/TS
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D03PKF Example Program Results'
      ITRACE = 0
      ITOL = 1
      ATOL(1) = 0.1D-3
      RTOL(1) = ATOL(1)
*
*     Set spatial-mesh points
*
      DO 20 I = 1, NPTS
         X(I) = (I-1.0D0)/(NPTS-1.0D0)
   20 CONTINUE
*
      XI(1) = 1.0D0
      NORM = 'A'
      LAOPT = 'F'
      IND = 0
      ITASK = 1
*
*     Set THETA to .TRUE. if the Theta integrator is required
*
      THETA = .FALSE.
      DO 40 I = 1, 30
         ALGOPT(I) = 0.0D0
   40 CONTINUE
      IF (THETA) THEN
         ALGOPT(1) = 2.0D0
      ELSE
         ALGOPT(1) = 0.0D0
      END IF
      ALGOPT(1) = 1.0D0
      ALGOPT(13) = 0.5D-2
*
*     Loop over output value of t
*
      TS = 1.0D-4
      TOUT = 0.0D0
*
      CALL UVINIT(NPDE,NPTS,X,U,NCODE,NEQN)
*
      DO 60 IT = 1, 5
         TOUT = 0.1D0*(2**IT)
```

```
         IFAIL = 1
*
         CALL D03PKF(NPDE,TS,TOUT,PDEDEF,BNDARY,U,NPTS,X,NLEFT,NCODE,
     +               ODEDEF,NXI,XI,NEQN,RTOL,ATOL,ITOL,NORM,LAOPT,
     +               ALGOPT,RSAVE,LRSAVE,ISAVE,LISAVE,ITASK,ITRACE,IND,
     +               IFAIL)
*
         IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99993) IFAIL
            GO TO 80
         ELSE IF (IT.EQ.1) THEN
            WRITE (NOUT,99997) ATOL, NPTS
            WRITE (NOUT,99999) X(1), X(5), X(9), X(13), X(21)
         END IF
*
*        Check against the exact solution
*
         CALL EXACT(TOUT,NEQN,NPTS,X,EXY)
*
         WRITE (NOUT,99998) TS
         WRITE (NOUT,99995) (U(I),I=1,25,8), U(41), U(43)
         WRITE (NOUT,99994) (EXY(I),I=1,25,8), EXY(41), TS
   60 CONTINUE
      WRITE (NOUT,99996) ISAVE(1), ISAVE(2), ISAVE(3), ISAVE(5)
*
   80 CONTINUE
*
99999 FORMAT ('  X        ',5F9.3,/)
99998 FORMAT (' T = ',F6.3)
99997 FORMAT (//'  Accuracy requirement =',E10.3,' Number of points = ',
     +       I3,/)
99996 FORMAT (' Number of integration steps in time = ',I6,/' Number o',
     +       'f function evaluations = ',I6,/' Number of Jacobian eval',
     +       'uations =',I6,/' Number of iterations = ',I6)
99995 FORMAT (1X,'App.  sol.  ',F7.3,4F9.3,'  ODE sol. =',F8.3)
99994 FORMAT (1X,'Exact sol.  ',F7.3,4F9.3,'  ODE sol. =',F8.3,/)
99993 FORMAT (1X,/1X,' ** D03PKF returned with IFAIL = ',I5)
      END
*
      SUBROUTINE UVINIT(NPDE,NPTS,X,U,NCODE,NEQN)
*     Routine for PDE initial values
*     .. Scalar Arguments ..
      INTEGER          NCODE, NEQN, NPDE, NPTS
*     .. Array Arguments ..
      DOUBLE PRECISION U(NEQN), X(NPTS)
*     .. Scalars in Common ..
      DOUBLE PRECISION TS
*     .. Local Scalars ..
      INTEGER          I, K
*     .. Intrinsic Functions ..
      INTRINSIC        EXP
*     .. Common blocks ..
      COMMON           /TAXIS/TS
*     .. Executable Statements ..
      K = 1
      DO 20 I = 1, NPTS
         U(K) = EXP(TS*(1.0D0-X(I))) - 1.0D0
         U(K+1) = -TS*EXP(TS*(1.0D0-X(I)))
         K = K + 2
   20 CONTINUE
      U(NEQN) = TS
      RETURN
      END
*
      SUBROUTINE ODEDEF(NPDE,T,NCODE,V,VDOT,NXI,XI,UCP,UCPX,UCPT,F,IRES)
*     .. Scalar Arguments ..
      DOUBLE PRECISION T
      INTEGER          IRES, NCODE, NPDE, NXI
*     .. Array Arguments ..
      DOUBLE PRECISION F(*), UCP(NPDE,*), UCPT(NPDE,*), UCPX(NPDE,*),
     +                 V(*), VDOT(*), XI(*)
```

```
*     .. Executable Statements ..
      IF (IRES.EQ.-1) THEN
         F(1) = VDOT(1)
      ELSE
         F(1) = VDOT(1) - V(1)*UCP(1,1) - UCP(2,1) - 1.0D0 - T
      END IF
      RETURN
      END
*
      SUBROUTINE PDEDEF(NPDE,T,X,U,UDOT,UX,NCODE,V,VDOT,RES,IRES)
*     .. Scalar Arguments ..
      DOUBLE PRECISION T, X
      INTEGER          IRES, NCODE, NPDE
*     .. Array Arguments ..
      DOUBLE PRECISION RES(NPDE), U(NPDE), UDOT(NPDE), UX(NPDE), V(*),
     +                 VDOT(*)
*     .. Executable Statements ..
      IF (IRES.EQ.-1) THEN
         RES(1) = V(1)*V(1)*UDOT(1) - X*U(2)*V(1)*VDOT(1)
         RES(2) = 0.0D0
      ELSE
         RES(1) = V(1)*V(1)*UDOT(1) - X*U(2)*V(1)*VDOT(1) - UX(2)
         RES(2) = U(2) - UX(1)
      END IF
      RETURN
      END
*
      SUBROUTINE BNDARY(NPDE,T,IBND,NOBC,U,UDOT,NCODE,V,VDOT,RES,IRES)
*     .. Scalar Arguments ..
      DOUBLE PRECISION T
      INTEGER          IBND, IRES, NCODE, NOBC, NPDE
*     .. Array Arguments ..
      DOUBLE PRECISION RES(NOBC), U(NPDE), UDOT(NPDE), V(*), VDOT(*)
*     .. Intrinsic Functions ..
      INTRINSIC        EXP
*     .. Executable Statements ..
      IF (IBND.EQ.0) THEN
         IF (IRES.EQ.-1) THEN
            RES(1) = 0.0D0
         ELSE
            RES(1) = U(2) + V(1)*EXP(T)
         END IF
      ELSE
         IF (IRES.EQ.-1) THEN
            RES(1) = V(1)*VDOT(1)
         ELSE
            RES(1) = U(2) + V(1)*VDOT(1)
         END IF
      END IF
      RETURN
      END
*
      SUBROUTINE EXACT(TIME,NEQN,NPTS,X,U)
*     Exact solution (for comparison purposes)
*     .. Scalar Arguments ..
      DOUBLE PRECISION TIME
      INTEGER          NEQN, NPTS
*     .. Array Arguments ..
      DOUBLE PRECISION U(NEQN), X(NPTS)
*     .. Local Scalars ..
      INTEGER          I, K
*     .. Intrinsic Functions ..
      INTRINSIC        EXP
*     .. Executable Statements ..
      K = 1
      DO 20 I = 1, NPTS
         U(K) = EXP(TIME*(1.0D0-X(I))) - 1.0D0
         K = K + 2
   20 CONTINUE
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D03PKF Example Program Results


  Accuracy requirement = 0.100E-03 Number of points =  21

 X            0.000    0.200    0.400    0.600    1.000

T =  0.200
App.  sol.    0.222    0.174    0.128    0.084    0.000  ODE sol. =   0.200
Exact sol.    0.221    0.174    0.127    0.083    0.000  ODE sol. =   0.200

T =  0.400
App.  sol.    0.492    0.377    0.271    0.174    0.000  ODE sol. =   0.400
Exact sol.    0.492    0.377    0.271    0.174    0.000  ODE sol. =   0.400

T =  0.800
App.  sol.    1.226    0.896    0.616    0.377    0.000  ODE sol. =   0.800
Exact sol.    1.226    0.896    0.616    0.377    0.000  ODE sol. =   0.800

T =  1.600
App.  sol.    3.952    2.595    1.610    0.895   -0.001  ODE sol. =   1.600
Exact sol.    3.953    2.597    1.612    0.896    0.000  ODE sol. =   1.600

T =  3.200
App.  sol.   23.522   11.918    5.807    2.588   -0.004  ODE sol. =   3.197
Exact sol.   23.533   11.936    5.821    2.597    0.000  ODE sol. =   3.200

Number of integration steps in time =    642
Number of function evaluations =   3022
Number of Jacobian evaluations =    39
Number of iterations =   1328
```

Example Program
Two PDEs Coupled with One ODE using Keller, Box and BDF
Solution $U(1,x,t)$




Two PDEs Coupled with One ODE using Keller, Box and BDF
Solution U(2,x,t)
U(2,x,t)
0
-10
-20
-30
-40
-50
-60
-70
-80
0.1
0.5
1
2
3
Time (logscale)
0
0.2
0.4
0.6
0.8
1
x